
// This WordPerfect macro makes it possible to search the text that is 
// selected in the New SOED application.
// It is possible to add a button to the toolbar.
// This can be done by selecting Preferences under the 
// File menu (WP 6.0) or the Edit menu(WP 6.1).
// Select the option "Button Bar"(WP 6.0) or "Toolbar" WP(6.1).
// Select a button bar and press "Edit". 
// Select "Play a macro" and press "Add Macro...".
// Select this file and press "Select".
// Close the Dialogs by Pressing Select, OK,Close and Close.
// A button should be on the button bar with the text "newsoed".

// Installation complete; the lines below are the macro instructions,
// please ignore these



// Macro Text:

// Try to load the DLL for external function calls
DLLLOAD(nLoadResult;"KERNEL")

// Check for errors
IF (nLoadResult<32)

GO (ErrorHandler)
ENDIF

// No Errors, so use the DLL

szServiceName=""
szPath=""
szExeName=""
szAppBaseName=""
szAppName=""
szBaseName=""

// Service Name
DLLCALL(nLoadResult; "GetPrivateProfileString"; nReturn: Word; {"Macro"; "ServiceName"; 
""; ADDRESS(ANSISTRING(szServiceName)); Loword(255);"newsoed.ini"})

// Path to exe
DLLCALL(nLoadResult; "GetPrivateProfileString"; nReturn: Word; {"Macro"; "Pathname"; ""; 
ADDRESS(ANSISTRING(szPath)); Loword(255);"newsoed.ini"})

// Executable name
DLLCALL(nLoadResult; "GetPrivateProfileString"; nReturn: Word; {"Macro"; "ExeName"; ""; 
ADDRESS(ANSISTRING(szBaseName)); Loword(255);"newsoed.ini"})

// Executable name
DLLCALL(nLoadResult; "GetPrivateProfileString"; nReturn: Word; {"Macro"; "AppName"; ""; 
ADDRESS(ANSISTRING(szAppBaseName)); Loword(255);"newsoed.ini"})

szExeName= szBaseName + ".EXE"
szAppName = szAppBaseName + "*"

IF (szServiceName="")
szServiceName := "NewSOED"
ENDIF

// Is Application running?
ASSIGN(hWnd; APPLOCATE(szAppName))
// If not start it



IF (hWnd = 0)
// Derive full path of application
szFullPath := szPath
IF (SUBSTR(szFullPath; STRLEN(szFullPath); 1) <> "\")

szFullPath := szFullPath + "\"
ENDIF

szFullPath := szFullPath+ szExeName

// Execute Application 
AppExecute(szFullPath)

// Check if Application is  started OK
IF  (APPLOCATE(szAppName) = 0)

GO (ErrorHandler)
ENDIF

ELSE
IF(hWnd) 

AppActivate(hWnd)
ENDIF

ENDIF

// Initiate DDE link
ddeID := 0
tries := 0

DDEInitiate(ddeID; szServiceName; "WP")

WHILE(ddeID = 0 AND tries < 5)
WAIT(10)
DDEInitiate(ddeID; szServiceName; "WP")
tries  := tries + 1
ENDWHILE

// Test for success
IF (ddeID = 0)

GO (ErrorHandler)
ENDIF

// What to send to Application
// If no selection select current word
IF (?SelectedTextSize = 0)

SelectWord()
ENDIF

IF (?SelectedTextSize > 0)
szSend := ?SelectedText



ELSE
szSend := " " // Use instead of Empty string to avoid WP Macro Error

ENDIF

// Send the string
DDEExecute (ddeID; szSend)
DDETerminate(ddeID)

// Release the DLL
DLLFREE(nLoadResult)

// End execution 
RETURN

// Error Handler
LABEL (ErrorHandler)

MESSAGEBOX(nReturn; "Error"; "Running the macro to search in the New Shorter 
Failed"; OK!|IconStop!|ApplicationModal!)

RETURN


